
La creación de un objeto puede ser tan simple como:

        oGraph := TGraph:New(nTtop,nLeft,oWnd,nWidth,nHeight)  // creación

        oGraph:AddSerie(aSerie,cLegend,nColor)                 // series

vía comando:

        @ nRow, nCol GRAPH oGraph OF oWnd SIZE nWidth, nHeight

        ADD SERIE TO oGraph SERIE aSerie LEGEND cLegend COLOR nColor

En recursos (Dll, RC):

        oGraph := TGraph:Redefine(nID,oDlg)

        oGraph:AddSerie(aSerie,cLegend,nColor)

vía comando:

        REDEFINE GRAPH oGraph ID nID OF oDlg

        ADD SERIE TO oGraph SERIE aSerie LEGEND cLegend COLOR nColor

        si el objeto se define vía comandos podrán indicarse las características

        básicas de la presentación (ver TGraph.Ch),

        REDEFINE GRAPH oGraph ID nID OF oDlg ;

           TITLE   "Mi Gráfica";    // título

           3D       ;               // despliegue en 3D

           XGRID    ;               // rejilla en X

           YGRID    ;               // rejilla en Y

           XVALUES  ;               // mostrar valores en X

           YVALUES  ;               // mostrar valores en Y

           LEGENDS  ;               // mostrar leyendas

           TYPE GRAPH_TYPE_BAR;     // tipo de gráfico a mostrar (BARRAS)

           POPUP    ;               // activar el menú de selección

Estas características así como aquellas que no pueden definirse vía comandos, pueden asignarse en forma directa:

        oGraph:lLegends := .F.   // para ocultar las leyendas

        oGraph:lViewVal := .T.   // para ver los valores de la serie

        oGraph:nXRanges := 10    // muestra 10 rangos en la rejilla

        oGraph:nBarD    := 20    // da mayor profundidad en 3D

        ...

Si deseamos que el usuario pueda jugar con la presentación del gráfico, 

puede ser suficiente con activar el menú POPUP, esto es:

        REDEFINE GRAPH oGraph ID nID OF oDlg POPUP

o bien

        oGraph:lPopUp := .T.

Esto permitirá que cuando el usuario presione el botón derecho del ratón,pueda obtener diferentes vistas del mismo gráfico, en forma muy simple. Todavía más, podemos habilitar o deshabilitar las opciones de 'Ver' y de selección de 'Fondo'.
        oGraph:lSelView := .F.   // deshalilita la opcion 'Ver' del menu

        oGraph:lSelBack := .T.   // permite colocar una imagen de fondo (Bmp)

Si observamos el código de la clase podremos ver las diferentes variables de instancia o características que pueden ser asignadas al objeto en cualquier momento:

   DATA    nType        // Tipo de gráfico           por definición  1-Barras

   DATA    nBarD        // Profundidad en efecto 3D                 15

   DATA    nBarSep      // Separador entre barras                    1

   DATA    l3D          // Efecto 3D                                .T.

   DATA    lxGrid       // Rejilla en X                             .T.

   DATA    lyGrid       // Rejilla en Y                             .T.

   DATA    lDotted      // Linea de rejilla separada o contínua     .T.

   DATA    lxVal        // Mostrar etiquetas en X                   .T.

   DATA    lyVal        // Mostrar etiquetas en Y                   .T.

   DATA    lTitle       // Mostrar títulos                          .T.

   DATA    lLegends     // Mostrar leyendas                         .T.

   DATA    cTitle       // Título                                   ''

   DATA    cSubTit      // Subtítulo                                ''

   DATA    cTitX        // Título en X                              ''

   DATA    cTitY        // Título en Y                              ''

   DATA    cPicture     // Máscara de valores                       '999,999.99'

   DATA    aFont        // Arreglo de fuentes (título y leyendas)

   DATA    nClr…        // Color de fuentes                         CLR_BLACK

   DATA    nXRanges     // Rangos en X                               4

   DATA    nValues      // Tipo de valores en gráfico de torta       1-series, 
                                                                     2-períodos

   DATA    lViewVal     // Mostrar valores de las series            .F.

   DATA    lBorders     // Bordes en gráfíco de barras              .F.

   DATA    lcTitle      // Centrar título                           .T.

   DATA    nClrGrid     // Color de la rejilla                      CLR_GRAY

   DATA    nClrBack     // Color de fondo                           CLR_WHITE

   DATA    nClrBLeg     // Fondo en leyendas                        CRL_WHITE
   DATA    cBitmap      // Imagen de fondo                          ''

   DATA    lPopUp       // Activar menú (PopUp)                     .F.

   DATA    lSelView     // Seleccionar opción 'Ver'                 .T.

   DATA    lSelBack     // Seleccionar opción 'Fondo'               .T.

   DATA    nLanguage    // Lengua (1-Inglés, 2-Español, 3-...)       1

   DATA    aSTitle      // Sombra en títulos                        .T.

Después de cambiar los valores de estas variables, o bien, de las series, será suficiente con refrescar el objeto

        oGraph:Refresh()

En cuanto a los métodos, practicamente todos son de uso interno a la propia clase, no obstante pueden usarse para obtener los valores de retorno, entre otros que sí son de utilidad se tiene:

   METHOD  AddSerie(aSerie,cLegend,nColor)       // para agregar series

   METHOD  SetYVals(aLabels)                     // para colocación de etiquetas

   METHOD  SelSerie(nSerie)                      // para seleccionar una serie

   METHOD  SelPeriod(nPeriod)                    // para seleccionar un periodo

   METHOD  Copy2ClipBoard()                      // envia imagen al portapapeles

   METHOD  Save2Bmp(cFile)                       // genera archivo bmp indicado

   METHOD  Print(oPrn,nTop,nLeft,nWidth,nHeight) // imprime el gráfico

---

Por algún motivo que no entiendo, la impresión no se logra en Windows 2000, NT y XP si alguien encuentra la forma de hacerlo favor de ponerlo a consideración de los compañeros del foro.

---

Los alcances dependen de la habilidad en el manejo de Clipper-FiveWin y de la imaginación del programador.

El test que acompaña a la Clase es una muestra de las posibilidades que pueden obtenerse con su uso.

alfredoarteaga@terra.com.mx

